

Master in Statistics and Operative Research

STOCHASTIC PROGRAMMING

Benders' Decomposition of a Stochastic Problem

Víctor Diví i Cuesta

UNIVERSITAT POLITÈCNICA DE CATALUNYA

Facultat de Matemàtiques i Estadística

Contents

| | |
|---------------------------------|---|
| 1. Introduction | 1 |
| 1.1. Problem | 1 |
| 1.2. Scenario | 2 |
| 2. Extensive Version | 3 |
| 3. Benders' Decomposition | 4 |
| 4. Conclusions | 6 |

List of Codes

| | |
|--|---|
| Code 1 AMPL Code for the Extensive model. | 3 |
| Code 2 AMPL Code for Benders' Decomposition models. | 4 |
| Code 3 AMPL Code for Benders' Decomposition. | 5 |

List of Tables

| | |
|--|---|
| Table 1 Evolution of costs for scenario l | 6 |
| Table 2 Evolution of costs for scenario u | 6 |

List of Summaries

| | |
|--|---|
| Summary 1 Results of the extensive model. | 3 |
|--|---|

1. Introduction

1.1. Problem

In the problem at hand, which we will call the ATM Deposit Problem, a bank branch wants to determine the amount of money to be deposited in an ATM on Fridays, before the weekend. The branch estimates that money has a cost (associated to the loss of benefits by interest rates) of c   for each   in the ATM. The demand of money during the weekend is a discrete random variable ξ , taking s values ξ_i with probabilities $p_i, i = 1, \dots, s$. The ATM has a capacity of u  , with a technical minimum of l  . If the demand is greater than x , then the ATM has to be refilled, with a cost of q   for each   the demand exceeds x . The bank branch formulates the following stochastic optimization problem in extensive form:

$$\begin{aligned}
 \text{Min} \quad & cx + \sum_{i=1}^s p_i q y_i \\
 \text{s.t.} \quad & l \leq x \leq u \\
 & x + y_i \geq \xi_i \quad i = 1, \dots, s \\
 & y_i \geq 0 \quad i = 1, \dots, s
 \end{aligned} \tag{P}$$

Fixing the value x to a valid value \bar{x} ($l \leq \bar{x} \leq u$) and removing the now-constant cost from the objective function, from the extensive version of the problem (P) we obtain the following problem:

$$\begin{aligned}
 \text{Min} \quad & \sum_{i=1}^s p_i q y_i \\
 \text{s.t.} \quad & y_i \geq \xi_i - \bar{x} \quad i = 1, \dots, s \\
 & y_i \geq 0 \quad i = 1, \dots, s
 \end{aligned} \tag{Q}$$

and its dual, which will be the Bender's Subproblem used:

$$\begin{aligned}
 \text{Max} \quad & \sum_{i=1}^s u_i (\xi_i - \bar{x}) \\
 \text{s.t.} \quad & u_i \leq p_i q \quad i = 1, \dots, s \\
 & u_i \geq 0 \quad i = 1, \dots, s
 \end{aligned} \tag{Q_D}$$

Using the Subproblem (Q_D), we rewrite the original extensive problem (P) as follows:

$$\begin{aligned}
 \text{Min} \quad & cx + \max \left\{ \sum_{i=1}^s u_i (\xi_i - x) : 0 \leq u_i \leq p_i q, i = 1, \dots, s \right\} \\
 \text{s.t.} \quad & l \leq x \leq u
 \end{aligned} \tag{P'}$$

With the set of vertices $\{v^1, \dots, v^V\}$ and extreme rays $\{r^1, \dots, r^R\}$ of the feasible polyhedron of (Q_D), we can rewrite (P') again as:

$$\begin{aligned}
 \text{Min} \quad & cx + z \\
 \text{s.t.} \quad & z \geq \sum_{j=1}^s v_j^i (d_j - x) \quad i = 1, \dots, V \\
 & 0 \geq \sum_{j=1}^s r_j^i (d_j - x) \quad i = 1, \dots, R \\
 & l \leq x \leq u
 \end{aligned} \tag{BP}$$

For the Benders' Decomposition, we will use a relaxation (BPr) of (BP) in which we only use a subset of vertices and extreme points:

$$\begin{aligned}
\text{Min} \quad & cx + z \\
\text{s.t.} \quad & z \geq \sum_{j=1}^s v_j^i (d_j - x) \quad i \in \mathcal{V} \subseteq \{1, \dots, V\} \\
& 0 \geq \sum_{j=1}^s r_j^i (d_j - x) \quad i \in \mathcal{R} \subseteq \{1, \dots, R\} \\
& l \leq x \leq u
\end{aligned} \tag{BPr}$$

However, note that (Q_D) is bounded, since all variables u_i have a lower bound (0) and an upper bound $(p_i q)$. In fact, it is straightforward to see that the values of u_i will be 0 if $\xi_i - \bar{x} \leq 0$ and $p_i q$ otherwise. Therefore, its feasible polyhedron doesn't have any extreme ray, so we can simplify (BPr) to obtain our **Master Problem** formulation:

$$\begin{aligned}
\text{Min} \quad & cx + z \\
\text{s.t.} \quad & z \geq \sum_{j=1}^s v_j^i (d_j - x) \quad i \in \mathcal{V} \subseteq \{1, \dots, V\} \\
& l \leq x \leq u
\end{aligned} \tag{MP}$$

1.2. Scenario

The scenario used in this report have the following characteristics:

$$\begin{aligned}
c &= 2.5 \cdot 10^{-4} \\
q &= 1.1 \cdot 10^{-3} \\
l &= 21 \cdot 10^3 \\
u &= 147 \cdot 10^3 \\
s &= 7
\end{aligned}$$

| i | p_i | ξ_i |
|-----|-------|---------|
| 1 | 0.04 | 150 |
| 2 | 0.09 | 120 |
| 3 | 0.10 | 110 |
| 4 | 0.21 | 100 |
| 5 | 0.27 | 80 |
| 6 | 0.23 | 60 |
| 7 | 0.06 | 50 |

The units of l , u and ξ_i are thousands of  .

The scenario data in **AMPL** format is provided in the **atm.dat** file.

2. Extensive Version

The model used to solve the extensive version of the ATM Problem is shown in Code 1. The code is also provided in the `extensive.mod` file, along a small runner script in the `extensive.run` file.

```

1 # Parameters
2 set S;                # Scenarios
3
4 param L ≥ 0;           # Minimum money to be deposited
5 param U > L;           # Maximum money to be deposited
6
7 param C;               # Cost of deposited money €/€
8 param Q;               # Cost of lack of money €/€
9
10 param P {S} ≥ 0 <= 1; # Probability of each scenario
11 param D {S} ≥ 0;       # Demand of each scenario
12
13 # Variables
14 var x ≥ L <= U;        # Amount deposited
15 var y {S} ≥ 0;         # Missing amount in each scenario
16
17 # Model
18 minimize Total_Cost:
19     C*x +
20     sum {i in S} P[i] * Q * y[i];
21
22 subject to Demand {i in S}:
23     x + y[i] ≥ D[i]
24
```

Code 1: AMPL Code for the Extensive model.

The results obtained with the execution of the extensive model can be seen in Summary 1. As can be seen, a total of 110k€ are deposited on the ATM, accounting for a cost of 27.5€. This amount falls short only in two scenarios, which adds 2.75€ to the expected cost for a total of 30.25€.

| Cost | Value |
|----------|-------|
| Deposit | 27.50 |
| Shortage | 2.75 |
| Total | 30.25 |

| Variable | x | y_1 | y_2 | y_3 | y_4 | y_5 | y_6 |
|----------|--------|-------|-------|-------|-------|-------|-------|
| Value | 110000 | 40000 | 10000 | 0 | 0 | 0 | 0 |

Summary 1: Results of the extensive model.

3. Benders' Decomposition

For the Benders' Decomposition, we use the formulations of the **Master Problem** and the **Subproblem** described in Section 1.1. The AMPL models for both are shown in Code 2, while the code for the actual Benders' Decomposition is shown in Code 3. The code is also provided in the `benders_simplified.mod` and `benders_simplified.run` files respectively.¹

```

1 # Common Params
2 set S;                                # Scenarios
3
4 param P {S} ≥ 0, ≤ 1;                 # Probability of each scenario
5 param D {S} ≥ 0;                      # Demand of each scenario
6
7
8 # Master Problem
9 param NCuts ≥ 0 integer;              # Number of cuts
10
11 param L ≥ 0;                          # Minimum money to be deposited
12 param U > L;                         # Maximum money to be deposited
13
14 param C;                             # Cost of deposited money €/€
15
16 param Y {S, 1..NCuts};               # Missing amount in each scenario
17
18 var x ≥ L, ≤ U;                      # Amount deposited
19 var z;                               # Maximum cost for missing money
20
21 minimize Total_Cost: C * x + z;
22
23 subject to Cuts {k in 1..NCuts}:
24     z ≥ sum {i in S} Y[i,k] * (D[i] - x);
25
26
27 # Subproblem
28 param Q;                             # Cost of lack of money €/€
29 param X;                             # Amount deposited
30
31 var u {S} ≥ 0;
32
33 maximize Dual_Cost: sum {i in S} u[i] * (D[i] - X);
34
35 subject to MissingCost {i in S}: u[i] ≤ P[i] * Q;

```

Code 2: AMPL Code for Benders' Decomposition models.

The AMPL code is a direct copy of the formulations presented above, with parameters in uppercase and variables in lowercase.

¹Another pair of AMPL files (`benders.mod` and `benders.run`) are provided without the simplification of the **Master Problem**, i.e. using (BPr) instead of (MP). The only difference is that it handles both point and ray cuts instead of only point cuts. In practice, both are equivalent since ray cuts are never generated.

```

1 reset;
2
3 model benders.mod
4 data atm.dat;
5
6 option solver cplex;
7
8 # Problem declaration (variables, Objective_Function, Restrictions)
9 problem Master: x, z, Total_Cost, Cuts;
10 problem Sub: u, Dual_Cost, MissingCost;
11
12 # Initializations
13 let NCuts := 0;
14 let X := L;
15 let z := C * X;
16
17 param GAP default Infinity;
18 param epsilon default 1.0e-8;
19
20 repeat {
21     solve Sub;
22     display Sub.result, Dual_Cost;
23
24     # Check gap
25     let GAP := abs(Dual_Cost - z);
26     display GAP;
27     if GAP <= epsilon * z then break;
28
29     # Add cut
30     let NCuts := NCuts + 1;
31     let {i in S} Y[i, NCuts] := u[i];
32
33     solve Master;
34     display Master.result, x, z, Total_Cost;
35
36     let X := x;
37 }
38
39 display x, Total_Cost;

```

Code 3: AMPL Code for Benders' Decomposition.

We start the algorithm initializing \bar{x} . Then, at each iteration, we start solving the **Subproblem** (which is guaranteed to find a solution). Then, we check if the gap between the dual cost and z is lower than the accepted relative tolerance εz (with a fixed $\varepsilon = 10^{-8}$). If it is, we stop. Otherwise, we create an optimality cut with the values of $u_i, i = 1, \dots, s$. We then proceed to solve the **Master Problem**, use the resulting x as the new value for \bar{x} , and move on to the next iteration.

Two scenarios have been executed, one in which we start with $\bar{x} = l$ and another in with $\bar{x} = u$. Both scenarios reach the same final results as the extensive model (shown in Summary 1). The evolution in the costs and \bar{x} over the iterations for each scenario are shown in Tables 1 and 2. The Gap column shows the absolute gap between the dual cost of the same iteration and the value of z of the previous one.

As can be seen, both scenarios end with the same objective value as the extensive version in the same number of iterations. In fact, note that the costs between scenarios only differ in the first two iterations. In the second execution of the **Master Problem** (in iteration 2), we obtain

the same result in both scenarios, and from that moment on, the results are always the same for both of them.

| Iteration | Dual Cost | Gap | Total Cost | z | x |
|-------------|-----------|-------------|------------|---------|---------|
| Init | - | - | - | 5.25 | 21000 |
| 1 | 72.82 | 67.57 | -29.03 | -65.78 | 147000 |
| 2 | 0.132 | 65.912 | 24.0242 | 2.87833 | 84583.3 |
| 3 | 12.7417 | 9.86333 | 28.642 | 1.892 | 107000 |
| 4 | 3.509 | 1.617 | 30.2358 | 1.55158 | 114737 |
| 5 | 2.07263 | 0.521053 | 30.25 | 2.75 | 110000 |
| 6 | 2.75 | 3.55271e-15 | - | - | - |

Table 1: Evolution of costs for scenario l .

| Iteration | Dual Cost | Gap | Total Cost | z | x |
|-------------|-----------|-------------|------------|---------|---------|
| Init | - | - | - | 36.75 | 147000 |
| 1 | 0.132 | 36.618 | 10.926 | 5.676 | 21000 |
| 2 | 72.82 | 67.144 | 24.0242 | 2.87833 | 84583.3 |
| 3 | 12.7417 | 9.86333 | 28.642 | 1.892 | 107000 |
| 4 | 3.509 | 1.617 | 30.2358 | 1.55158 | 114737 |
| 5 | 2.07263 | 0.521053 | 30.25 | 2.75 | 110000 |
| 6 | 2.75 | 3.55271e-15 | - | - | - |

Table 2: Evolution of costs for scenario u .

4. Conclusions

In this report we have solved an scenario of the stochastic ATM Deposit Problem using two different methods: solving directly the classical MILP formulation and using the Benders' Decomposition (with two different initial scenarios). We have presented a classic Benders' Master and subproblem formulations, as well as a simplification of the Master formulation that relies on the nature of the problem at hand. The computational results show that both methods (extensive version and Bender's Decomposition) yield exactly the same results.