

Master in Statistics and Operative Research

LARGE SCALE OPTIMIZATION

# Solving a Network Design Problem using Benders' Decomposition

*Víctor Diví i Cuesta*

UNIVERSITAT POLITÈCNICA DE CATALUNYA

Facultat de Matemàtiques i Estadística

## Contents

1. Introduction .....	1
1.1. Problem .....	1
1.2. Scenario .....	2
2. Extensive Version .....	3
3. Benders' Decomposition .....	5
4. Conclusions .....	11

## List of Codes

Code 1 AMPL Code for Extensive model. ....	4
Code 2 AMPL Code for Benders' Decomposition models. ....	6
Code 3 AMPL Code for Benders' Decomposition algorithm. ....	7

## List of Tables

Table 1 Evolution of costs for scenario 0. ....	8
Table 2 Evolution of costs for scenario 1. ....	8
Table 3 Differences in link flows between solutions for origin $l = 3$ .....	11
Table 4 Differences in link flows between solutions for origin $l = 11$ .....	11

## List of Summaries

Summary 1 Scenario 19. ....	3
Summary 2 Results of the extensive model. ....	5
Summary 3 Results of the Scenario 0. ....	9
Summary 4 Results of the Scenario 1. ....	10

# 1. Introduction

## 1.1. Problem

In the Network Design Problem at hand, we initially have a network configuration given by  $G' = (N, A')$ , with the possibility to extend it taking candidate links from a set  $\hat{A}$ , so that the final network may become  $G = (N, A)$  with  $A = A' \cup \hat{A}$ . Within the network, we have to distribute a certain amount of flow from a set of origin nodes  $O \subseteq N$  to a set of destination nodes  $D \subseteq N$ . Each destination  $d \in D$  requires a fixed amount of flow from each origin  $l \in O$ . Links between nodes have infinite capacity.

The problem is identifying the suitable subset  $\hat{A}' \subseteq \hat{A}$  so that adding them to the network minimizes the total cost, which is made up by two components:

a) Investment cost: fixed cost for adding a new link from  $\hat{A}$  to the network.

$$F_{ij}, \forall (i, j) \in \hat{A}$$

b) Exploitation cost: origin-dependent cost per unit of flow crossing a link in the network.

$$C_{ij}^l, \forall (i, j) \in A, \forall l \in O$$

Let  $g_d^l$  be the demand required by node  $d \in D$  from node  $l \in O$ , we define  $T_i^l$  as:

$$T_i^l = \begin{cases} \sum_{d \in D} g_d^l, & \text{if } i = l \\ -g_i^l, & \text{if } i \in D \\ 0, & \text{otherwise} \end{cases} \quad \forall i \in N, \forall l \in O$$

Then, with variables  $x_{ij}^l$  representing the flow from origin  $l \in O$  crossing arc  $(i, j) \in A \cup \hat{A}$  and  $y_{ij}$  representing whether or not arc  $(i, j) \in \hat{A}$  is built, we have the following problem:

$$\begin{aligned} \text{Min} \quad & \sum_{a \in \hat{A}} F_a y_a + \sum_{l \in O} \sum_{a \in A} C_a^l x_a^l \\ \text{s.t.} \quad & \sum_{(i,j) \in A} x_{ij}^l - \sum_{(j,i) \in A} x_{ji}^l = T_i^l \quad \forall i \in N, \forall l \in O \\ & x_a^l \leq \rho y_a \quad \forall a \in \hat{A}, \forall l \in O \\ & x_a^l \geq 0 \quad \forall a \in A, \forall l \in O \\ & y_a \in \{0, 1\} \quad \forall a \in \hat{A} \end{aligned}$$

Constant  $\rho$  must be big enough so that it doesn't interfere with the arc flows ( $\rho > \sum_{l \in O} T_l^l$ ).

To solve the problem using Benders' Decomposition, we won't solve the problem directly, but rather iteratively solve a **Master Problem** and a **SubProblem** until the values of both converge. For a fixed solution  $\bar{y}$  for variables  $y$ , the structure of the **SubProblem** is as follows:

$$\begin{aligned} \text{Min} \quad & \sum_{a \in \hat{A}} F_a \bar{y}_a + \sum_{l \in O} \sum_{a \in A} C_a^l x_a^l \\ \text{s.t.} \quad & \sum_{(i,j) \in A} x_{ij}^l - \sum_{(j,i) \in A} x_{ji}^l = T_i^l \quad \forall i \in N, \forall l \in O \tag{A} \\ & x_a^l \leq \rho \bar{y}_a \quad \forall a \in \hat{A}, \forall l \in O \tag{B} \\ & x_a^l \geq 0 \quad \forall a \in A, \forall l \in O \tag{C} \end{aligned}$$

Meanwhile, at iteration M of the algorithm, the **Master Problem** has the following structure:

$$\begin{aligned}
\text{Min} \quad & \sum_{a \in \hat{A}} F_a y_a + z \\
\text{s.t.} \quad & z \geq \sum_{l \in O} \left( \sum_{i \in N} T_i^l U_i^{l^k} - \rho \sum_{\substack{a \in \hat{A} \\ \bar{y}_a^k = 1}} \hat{\tau}_a^{l^k} y_a \right) \quad \forall k = 1, \dots, M \\
& y_a \in \{0, 1\} \quad \forall a \in \hat{A}
\end{aligned}$$

Where  $\bar{y}^k$  is the value of  $\bar{y}$  at iteration  $k$ , and  $U^k$  and  $\hat{\tau}^k$  are the Lagrangian multipliers of (A) and (B) respectively at iteration  $k$ , with  $U^k$  being equal to the dual variables of (A) and  $\tau^k$  is computed as follows (skipping super-index  $k$  for  $\tau$  and  $U$ ):

$$\tau_{ij}^l = \max(0, U_i^l - U_j^l - C_{ij}^l) \quad \forall (ij) \in \hat{A}$$

Note that the formulation used in this report is slightly different (although equivalent) than the one presented in the questionnaire. Apart from notation changes done to be closer to the implementation, the main big changes are:

- **SubProblem** cost includes investment cost of  $\bar{y}$ , so  $z_D$  is the objective value of the **SubProblem**, instead of being the fixed cost plus the value of the **SubProblem**.
- Investment cost in the **Master Problem** is moved from each cut to the objective function, since it depends in the current value of  $y$  and therefore equal in all cuts.

## 1.2. Scenario


The scenario used in this report is Scenario 19, with:

$$\begin{aligned}
N &= \{1, \dots, 24\} \\
O &= \{3, 11\} \\
D &= \{5, 8, 12, 13, 14, 17, 18, 19, 21, 23\} \\
F_{ij} &= 10|x_i - x_j| + 60|y_i - y_j| \quad \forall (i, j) \in \hat{A} \\
C_{ij}^l &= 95 + (x_i - x_j)^2 + 8(y_i - y_j)^2 \quad \forall (i, j) \in A \cup \hat{A}, \forall l \in O \\
T_i^l &= \begin{cases} 500, & \text{if } i = l \\ -50, & \text{if } i \in D \\ 0, & \text{otherwise} \end{cases} \quad \forall i \in N, \forall l \in O
\end{aligned}$$

The costs for candidate links and the network graph of the scenario can be seen in Summary 1. The origins ( $\{3, 11\}$ ) are represented with colored double-circled nodes, while their corresponding destinations are represented with nodes filled with their color. In this case, both origins have the same destinations, so all of them are colored with both colors. Preexisting links are represented with solid edges, while dashed ones represent the candidate links that can be added to the network.

The scenario data in AMPL format is provided in the `network_design.dat` file.

Arc	Invest- ment Cost	Exploita- tion Cost
1 $\rightarrow$ 2	10	96
2 $\rightarrow$ 3	10	96
4 $\rightarrow$ 5	10	96
10 $\rightarrow$ 5	60	103
14 $\rightarrow$ 13	10	96
15 $\rightarrow$ 13	103.65	25
16 $\rightarrow$ 15	103.65	25
19 $\rightarrow$ 7	240	223
21 $\rightarrow$ 18	60	103
22 $\rightarrow$ 23	20	99



Summary 1: Scenario 19. Origins are represented by double-circled nodes, while destinations by filled nodes. Candidate links are represented by dashed edges. Table only shows costs for candidate links.

## 2. Extensive Version

The model used to solve the extensive version of the Network Design Problem is shown in Code 1. The code is also provided in the `extensive.mod` file, along a small runner script in the `extensive.run` file.

---

```

1 # Parameters
2 set N;
3 set O within N;
4 set A within N cross N;
5 set Ahat within N cross N;
6 set AA := A union Ahat;
7
8 param XC {N};
9 param YC {N};
10 param T {i in N, l in 0};
11 param C {(i,j) in AA, l in 0} :=
12     95 + (XC[i] - XC[j])^2
13     + 8 * (YC[i] - YC[j])^2;
14 param F {(i,j) in Ahat} :=
15     10 * abs(XC[i] - XC[j]) +
16     60 * abs(YC[i] - YC[j]);
17 param RHO > 0;
18
19 # Variables
20 var y {(i,j) in Ahat} binary;
21
22 # Flow variables and constraints
23 node I {i in N, l in 0}:
24     net_out = T[i,l];
25 arc x_l {(i,j) in AA, l in 0} >= 0:
26     from I [i,l],
27     to I [j,l];
28
29 # Model
30 # Total cost
31 minimize Total_Cost:
32     sum {(i,j) in Ahat} F[i,j] * y[i,j] +
33     sum {l in 0, (i,j) in AA} C[i,j,l] * x_l[i,j,l];
34
35 # Build-to-use Constraint
36 subject to Capacity {(i,j) in Ahat, l in 0}:
37     x_l[i,j,l] <= RHO * y[i,j];

```

---

Code 1: AMPL Code for Extensive model.


The results obtained with the execution of the extensive model can be seen in Summary 2. As can be seen, a total of three candidate links are added with an investment cost ( $f^\top y$ ) of 40:

$$\hat{A}_1 = \{a \mid a \in \hat{A}, y_a = 1\} = \{(4, 5), (14, 13), (22, 23)\}$$

The exploitation costs add up to 346200 ( $\sum_{l \in O} \sum_{a \in A} C_a^l x_a^l$ ), for a total cost  $\bar{z}$  of 346240.

The flows of the final solution are listed in the flow tables of Summary 2, and which links are used can be seen in the visualization of the network, where edges are painted according to the flow they carry from each origin. Candidate links can be identified by dashed edges.

The execution time of the `solve` command is approximately 0.5 seconds, although measurements are not very consistent.



Summary 2: Results of the extensive model. Candidate links are represented by dashed edges in the graph and green rows in the flow tables. Unused links are greyed in the graph and not included in the flow tables.

### 3. Benders' Decomposition

For the Benders' Decomposition Algorithm, we start by transforming the original problem into two different problems: the **Master Problem** and the **SubProblem**. The AMPL models for both are shown in Code 2, while the code for the actual Benders' Decomposition is shown in Code 3. The code is also provided in the `benders.mod` and `benders.run` files respectively.

Two scenarios have been executed, one in which we start adding none of the candidate links (Scenario 0:  $y_a = 0, \forall a \in \hat{A}$ ) and another in which we start adding all of them (Scenario 1:  $y_a = 1, \forall a \in \hat{A}$ ). For each scenario, a table describing the evolution of the costs throughout the iterations (Table 1 and Table 2) and a summary showing the final arc flows (Summary 3 and Summary 4) are shown.

---

```

1 # Common Parameters
2 set N; # Nodes
3 set O within N; # Origins
4 set A within N cross N; # Existing Arcs
5 set Ahat within N cross N; # Potential new Arcs
6 set AA:=A union Ahat; # All Arcs
7
8 param XC {N}; # Node X Coordinate
9 param YC {N}; # Node Y Coordinate
10 param T {i in N, l in O}; # Out Flow
11 param RHO > 0; # Maximum arc capacity
12
13 # Arc costs
14 param C {(i,j) in AA, l in O} := # Exploitation Cost
15     95 + (XC[i] - XC[j])^2 +
16     8 * (YC[i] - YC[j])^2;
17 param F {(i,j) in Ahat} := # Investment Cost
18     10 * abs(XC[i] - XC[j]) +
19     60 * abs(YC[i] - YC[j]);
20
21 # -----
22 # Subproblem
23 # -----
24 param Y {(i,j) in Ahat};
25
26 # Flow restrictions and variables
27 node Node_Constraints {i in N, l in O}:
28     net_out = T[i,l];
29 arc x_l {(i,j) in AA, l in O} >= 0:
30     from Node_Constraints [i,l],
31     to Node_Constraints [j,l];
32
33 minimize SubProblem_Cost:
34     sum {(i,j) in Ahat} F[i, j] * Y[i, j] +
35     sum {(i,j) in AA, l in O} C[i,j,l] * x_l[i,j,l];
36
37 subject to Build_To_Use_Constraints {(i,j) in Ahat, l in O}:
38     x_l[i,j,l] <= RHO * Y[i,j];
39
40
41 # -----
42 # Master problem
43 # -----
44 param NCuts; # Number of cuts
45 param YK {(i,j) in Ahat, k in 1..NCuts}; # Arc constructed in iteration k
46 param Cut {(i,j) in Ahat, l in O, k in 1..NCuts};
47 param U {i in N, l in O, k in 1..NCuts};
48
49 var y {(i,j) in Ahat} binary; # Whether arc is built or not
50 var z >= 0; # Usage cost
51
52 minimize Total_Cost: sum {(i, j) in Ahat} (F[i, j] * y[i, j]) + z;
53
54 subject to Cuts {k in 1..NCuts}:
55     z >=
56     sum {l in O} (
57         sum {i in N} T[i, l] * U[i, l, k]
58         - RHO * sum {(i, j) in Ahat: YK[i, j, k] = 0} Cut[i, j, l, k] * y[i, j]
59     )
60 ;

```

---

Code 2: AMPL Code for Benders' Decomposition models.



---

```

1 model benders.mod;
2 data network_design.dat;
3 option solver cplex;
4
5 # Problem definition (Variables, Objective, Constraints)
6 problem SUBPR: xl, SubProblem_Cost, Build_To_Use_Constraints, Node_Constraints;
7 problem MASTER: z, y, Total_Cost, Cuts;
8
9 # Initializations
10 let NCuts := 0;
11 let {(i,j) in Ahat} Y[i,j] := 0; # 0 or 1
12 param GAP;
13 param epsilon = 1.0e-8;
14
15 # Benders' Iterations
16 repeat {
17     # RESOLVE SUBPROBLEM
18     solve SUBPR;
19     display SUBPR.result, SubProblem_Cost;
20
21     let NCuts := NCuts + 1;
22
23     let {i in N, l in 0} U[i, l, NCuts] := -Node_Constraints.dual[i, l];
24     let {(i,j) in Ahat, l in 0} Cut[i, j, l, NCuts] :=
25         max(0, U[i, l, NCuts] - U[j, l, NCuts] - C[i, j, l]);
26
27     let {(i, j) in Ahat} YK[i, j, NCuts] := Y[i,j];
28
29     # RESOLVE MASTER PROBLEM
30     solve MASTER;
31     display MASTER.result, y;
32
33     let GAP := abs(SubProblem_Cost - Total_Cost);
34     display GAP;
35     if GAP <= epsilon * Total_Cost then break;
36
37     let {(i,j) in Ahat} Y[i,j] := y[i,j];
38 }
39 display Total_Cost, y;

```

---

Code 3: AMPL Code for Benders' Decomposition algorithm.

As can be seen in the tables, both scenarios end with the same objective value as the extensive version and with a similar number of iterations (11 for Scenario 0 and 10 for Scenario 1).

Note that even though both scenarios have the same objective value and the same candidate links are added, the flows in the final configuration all differ from each other. These differences are clear in the visualizations provided in the summaries. For instance, the solution obtained in Scenario 0 using the Benders' Decomposition (Summary 3), there are two arcs ((13, 17) and (14, 18)) that are used for both origins (shown in purple in the visualization), while this doesn't happen for neither of the other scenarios (Summary 2 and Summary 4).

These differences are shown in detail in Tables 3 and 4, where the links with differences in their flows and their associated cost are listed. As can be seen, even with the differences in the flows, the resulting costs for these arcs are the same in all solutions. Note that links with the same flow in all solutions are not listed.

The execution time of all the `solve` commands executed add up to approximately 0.1 seconds, although similarly to the extensive model, this measurement is not consistent.


Iteration	$z_D$	Investment Cost	Exploitation Cost	$\bar{z}$
1	369950	0	369950	10
2	369960	10	369950	100
3	346300	100	346200	336220
4	364320	20	364300	336220
5	362070	20	362050	336230
6	356430	30	356400	336230
7	359780	30	359750	336240
8	354140	40	354100	336240
9	351890	40	351850	336250
10	346250	50	346200	346240
11	346240	40	346200	346240

Table 1: Evolution of costs for scenario 0.

Iteration	$z_D$	Investment Cost	Exploitation Cost	$\bar{z}$
1	346750	550	346200	346200
2	369950	0	369950	346210
3	369960	10	369950	346210
4	364310	10	364300	346210
5	362060	10	362050	346220
6	356420	20	356400	346220
7	359770	20	359750	346230
8	354130	30	354100	346230
9	351880	30	351850	346240
10	346240	40	346200	346240

Table 2: Evolution of costs for scenario 1.





Summary 4: Results of Benders' Decomposition for Scenario 1. Candidate links are represented by dashed edges in the graph and green rows in the flow tables. Unused links are greyed in the graph and not included in the flow tables.

Differences for $l = 3$						
Arc	Extensive		Benders 0		Benders 1	
	Flow	Cost	Flow	Cost	Flow	Cost
1 $\rightarrow$ 4	50	5150	0	0	0	0
10 $\rightarrow$ 9	0	0	50	4800	50	4800
13 $\rightarrow$ 17	0	0	50	6350	0	0
14 $\rightarrow$ 13	100	9600	150	14400	100	9600
14 $\rightarrow$ 18	150	19050	100	12700	150	19050
18 $\rightarrow$ 17	50	4800	0	0	50	4800
2 $\rightarrow$ 1	50	4800	0	0	0	0
3 $\rightarrow$ 2	50	4800	0	0	0	0
3 $\rightarrow$ 6	350	36050	400	41200	400	41200
4 $\rightarrow$ 9	50	5150	0	0	0	0
5 $\rightarrow$ 10	0	0	50	5150	50	5150
6 $\rightarrow$ 5	50	4800	100	9600	100	9600
<b>Total</b>	<b>94200</b>		<b>94200</b>		<b>94200</b>	

Table 3: Differences in link flows between solutions for origin  $l = 3$ 

Differences for $l = 11$						
Arc	Extensive		Benders 0		Benders 1	
	Flow	Cost	Flow	Cost	Flow	Cost
13 $\rightarrow$ 14	50	4800	100	9600	50	4800
13 $\rightarrow$ 17	100	12700	50	6350	100	12700
14 $\rightarrow$ 18	0	0	50	6350	0	0
17 $\rightarrow$ 18	50	4800	0	0	50	4800
<b>Total</b>	<b>22300</b>		<b>22300</b>		<b>22300</b>	

Table 4: Differences in link flows between solutions for origin  $l = 11$ 

## 4. Conclusions

In this report we have solved a scenario of a Network Design Problem using two different methods: solving directly the classical MILP formulation and using the Benders' Decomposition (with two different initial scenarios). Comparing the results obtained, we see that both of them reach the optimal cost, although the actual solutions generated differ from each other. Time-wise, and although measurements across executions of the same method are not consistent, the Benders' Decomposition does consistently outperform the Extensive model, even with a small scenario as the one we deal with.